A little jaunt brought on by a seemingly reasonable claim

... the premise behind astrology is ridiculous ... the doctor or midwife or taxi driver who helped deliver you exerted a greater gravitational pull on you at your moment of birth than did the sun, the moon, or any of the planets.

from *The Canon: A Whirligig Tour of the Beautiful Basics of Science* by Natalie Angier

Let us calculate.

The gravitational force between two objects varies with their masses and inversely with the square of the distance between them

In[1]:= $F[m1\_, m2\_, r\_] := G \frac{m1\ m2}{r^2}$

We'll need the gravitational constant

In[2]:= `<< PhysicalConstants``

In[3]:= `G = GravitationalConstant`

Out[3]= $\frac{6.673 \times 10^{-11}\ \text{Meter}^2\ \text{Newton}}{\text{Kilogram}^2}$

While the quote addresses a person and a baby, we'll consider two adults, each weighing, say, 100 kg. If they're 1/2 meter apart, the force between them is

In[4]:= $F_{twopeople} = F\left[100\ \text{Kilogram},\ 100\ \text{Kilogram},\ \frac{1}{2}\ \text{Meter}\right]$

Out[4]= 2.6692×10^{-6} Newton

which is actually pretty large. For instance, this is equivalent to an amount of, say, salt, of mass

In[5]:= $m_{salt} := \text{Convert}\left[\frac{F_{twopeople}}{\text{AccelerationDueToGravity}},\ \text{Kilogram}\right];\ m_{salt}$

Out[5]= 2.72183×10^{-7} Kilogram

The density of salt is about $\rho = 2150\, kg / m^3$ so from $m = \rho V$, the equivalent volume is

In[6]:= $V_{salt} := \frac{m_{salt}}{2150\ \frac{\text{Kilogram}}{\text{Meter}^3}};\ V_{salt}$

Out[6]= $1.26597 \times 10^{-10}\ \text{Meter}^3$

or, since there are 1,000 millimeters per meter, 0.126 mm^3, which is a cube about 1/2 mm on a side.

Which means that the person sitting across the table from you is exerting a gravitation force on you equivalent to the weight of a grain or two of salt in the palm of your hand.

So, back to the planets. First, the Earth is clearly a planet and also clearly exerts quite a bit of force on you, far more than any taxi driver or midwife. But ignoring that, what about the other celestial bodies?

Let's consider Jupiter. Jupiter is pretty big, but it's also pretty far away, so just knowing the form of the force law won't let us figure out if the force between you and Jupiter is large (we've got two competing effects, $F \sim m_{\text{jup}}$ and $F \sim {r_{\text{jup}}}^{-2}$, one is big for Jupiter and one little). We'll need the actual numbers. Jupiter's mass is easy to get

```
In[7]:= m_jup = AstronomicalData["Jupiter", "Mass"] Kilogram
```

```
Out[7]= 1.8988 × 10^27 Kilogram
```

But how far away is it? This value is changing all the time, of course, but we can get a good idea of what it is by calculating the maximum, so whatever result we get, we'll know that the real answer is at least that but almost always more.

The maximum occurs (approximately, since Jupiter and Earth move around the Sun in ellipses and not circles) when Jupiter is directly opposed to Earth, i.e. directly opposit with the Sun in the middle. We can estimate how far this is by taking the sum of the maximum distance of both Jupiter and the Earth from the Sun

```
In[8]:= r_jup = (AstronomicalData["Jupiter", "Apoapsis"] + AstronomicalData["Earth", "Apoapsis"]) Meter
```

```
Out[8]= 9.6817916 × 10^11 Meter
```

So what's the force between you and Jupiter?

```
In[9]:= F_jup = F[100 Kilogram, m_jup, r_jup]
```

```
Out[9]= 0.0000135173 Newton
```

Comparing to the force between you and a taxi driver

In[10]:= $\dfrac{F_{\text{jup}}}{F_{\text{twopeople}}}$

```
Out[10]= 5.06416
```

Oops! Jupiter's force is five times as large. And since we took the largest distance, this is the lower limit.

This raises the issue of what forces the other heavenly bodies exert on you. First, though, let's simplify the math we have to do.

We're eventually going to be taking a ratio of two forces and some things are common to both numerator and denominator. Specifically, G appears in both as does the mass of the person, so these both cancel, leaving

In[11]:= Fratio[m2_, m3_, r12_, r13_] := $\dfrac{\text{m2}}{\text{m3}}\,\dfrac{\text{r13}^2}{\text{r12}^2}$

Just to check, we can calculate the answer for Jupiter and a taxi driver again

In[12]:= Fratio[m_{jup}, 100 Kilogram, r_{jup}, $\frac{1}{2}$ Meter]

```
Out[12]= 5.0642
```

Now we need the masses and greatest distances to the varous planets.

Mathematica reports the distance between the Moon and Earth in kilometers and all other Apoapses is meters, so we have to fix that below. We can also use the radius of the Earth for r_{earth} to get the relative force between it and a taxi driver.

In[13]:=
```
celestialbodynames = Append[AstronomicalData["Planet"], "Moon"]
celestialbodies = {AstronomicalData[ToString[#], "Mass"] Kilogram,
     Abs[AstronomicalData[ToString[#], "Apoapsis"] +
        AstronomicalData[ToString["Earth"], "Apoapsis"]] Meter} & /@ celestialbodynames;
celestialbodies[[3, 2]] = AstronomicalData["Earth", "Radius"] Meter;
celestialbodies[[10, 2]] = 1000 AstronomicalData["Moon", "Apoapsis"] Meter;
celestialbodies
```

Out[13]= {Mercury, Venus, Earth, Mars, Jupiter, Saturn, Uranus, Neptune, Pluto, Moon}

Out[17]= $\{\{3.3022\times10^{23}$ Kilogram, 2.21914780×10^{11} Meter$\}$, $\{4.8690\times10^{24}$ Kilogram, 2.6103955×10^{11} Meter$\}$, $\{5.9742\times10^{24}$ Kilogram, 6.378140×10^{6} Meter$\}$, $\{6.4191\times10^{23}$ Kilogram, 4.01326430×10^{11} Meter$\}$, $\{1.8988\times10^{27}$ Kilogram, 9.6817916×10^{11} Meter$\}$, $\{5.6850\times10^{26}$ Kilogram, 1.6560812×10^{12} Meter$\}$, $\{8.6625\times10^{25}$ Kilogram, 3.15848711×10^{12} Meter$\}$, $\{1.0278\times10^{26}$ Kilogram, 4.6889720×10^{12} Meter$\}$, $\{1.314\times10^{22}$ Kilogram, 7.5280256×10^{12} Meter$\}$, $\{7.3459\times10^{22}$ Kilogram, 4.057×10^{8} Meter$\}\}$

So now all we need to do is calculate the forces.

In[18]:= ratios = $\left(\text{Fratio}\left[\#[[1]], 100\ \text{Kilogram}, \#[[2]], \frac{1}{2}\ \text{Meter}\right]\right)$ & /@ celestialbodies

Out[18]= $\{0.016764, 0.17864, 3.6714\times10^{8}, 0.0099637, 5.0642, 0.51821, 0.021708, 0.011687, 5.797\times10^{-7}, 1116.\}$

For the curious, this line uses what *Mathematica* calles a Pure Function (the (#) & bit) which is applied to each pair of masses and distances in turn (the /@ bit).

So, our answer in a more convenient form:

In[19]:=
```
{celestialbodynames, ratios} // Transpose // MatrixForm
```

Out[19]//MatrixForm=

Mercury	0.016764
Venus	0.17864
Earth	3.6714×10^{8}
Mars	0.0099637
Jupiter	5.0642
Saturn	0.51821
Uranus	0.021708
Neptune	0.011687
Pluto	5.797×10^{-7}
Moon	1116.

Looks like the Earth, the Moon, and Jupiter all exert more force than your average taxi driver. But the taxi driver clearly 'out gravity's everything else.

Though, we seem to be missing a rather large heavenly body

In[20]:=
```
Fratio[AstronomicalData["Sun", "Mass"] Kilogram,
  100 Kilogram, AstronomicalData["Earth", "Apoapsis"] Meter, 1/2 Meter]
```

Out[20]= 214 885.0

Bigger than anything but the Earth itself.

Which raises another issue, why does the Moon contribute more to the ocean tides than the Sun? The Sun's gravitational force on the water in the ocean is ~100 times larger than the Moon's.

I leave that as an exercise for the reader. A hint: considering just the Earth and the Moon, do two identical rocks on opposit sides of the planet always have the same net 'weight'?

But before we go ...

It's frequently a good idea to see how the answers change as you vary the conditions somewhat. So let's consider what happens when we're as *close* as possible to the planet, instead of as far away as possible.

We'll just change the + to a –

In[21]:=
```
celestialbodies = {AstronomicalData[ToString[#], "Mass"] Kilogram,
     Abs[AstronomicalData[ToString[#], "Apoapsis"] -
        AstronomicalData[ToString["Earth"], "Apoapsis"]] Meter} & /@ celestialbodynames;
celestialbodies[[3, 2]] = AstronomicalData["Earth", "Radius"] Meter;
celestialbodies[[10, 2]] = 1000 AstronomicalData["Moon", "Apoapsis"] Meter;
ratiosclose = (Fratio[#[[1]], 100 Kilogram, #[[2]], 1/2 Meter]) & /@ celestialbodies;
{celestialbodynames, ratiosclose} // Transpose // MatrixForm
```

Out[25]//MatrixForm=

$$\begin{pmatrix} \text{Mercury} & 0.12194 \\ \text{Venus} & 6.5358 \\ \text{Earth} & 3.6714\times10^{8} \\ \text{Mars} & 0.17010 \\ \text{Jupiter} & 10.767 \\ \text{Saturn} & 0.77766 \\ \text{Uranus} & 0.026582 \\ \text{Neptune} & 0.013365 \\ \text{Pluto} & 6.295\times10^{-7} \\ \text{Moon} & 1116. \end{pmatrix}$$

Big difference. Venus now wins over the taxi driver and Jupiter doubled its effect. And look at the relative change of Mars

In[26]:=
```
{celestialbodynames, ratiosclose / ratios} // Transpose // MatrixForm
```

Out[26]//MatrixForm=

$$\begin{pmatrix} \text{Mercury} & 7.274 \\ \text{Venus} & 36.588 \\ \text{Earth} & 1.0000 \\ \text{Mars} & 17.072 \\ \text{Jupiter} & 2.1262 \\ \text{Saturn} & 1.5007 \\ \text{Uranus} & 1.2245 \\ \text{Neptune} & 1.1436 \\ \text{Pluto} & 1.086 \\ \text{Moon} & 1.000 \end{pmatrix}$$

We can conclude from this that Mercury's, Venus', and Mar's gravitational effect is dominated by its distance from us, while the force due to the larger planets depends mostly on their mass.

But we're not done yet. Other things that vary a lot are the mass of the person and their distance from us. Looking at the equation, we can easily see that a 50kg (110lb) person will exert half the force. But what about, say, the taxi driver's hands holding a new born baby? Here the distance is quite a bit closer. Will that change anything?

I think a good estimate of the mass of a person's hand is about a pound, so two hands is about a kilogram. What of the distance? Before we could be cavalier about that since the planets are so far away, any little change we make at human scales is irrelevent to the final answer. Now however, we have to be more careful. The distance isn't zero, since the distance in the force equation is the distance between two centers of mass. For a baby's head that's about 5cm from the surface and for your hand about another centimeter, for six centimeters total.

In[27]:=
```
celestialbodynames = Append[AstronomicalData["Planet"], "Moon"]
celestialbodies = {AstronomicalData[ToString[#], "Mass"] Kilogram,
     Abs[AstronomicalData[ToString[#], "Apoapsis"] +
        AstronomicalData[ToString["Earth"], "Apoapsis"]] Meter} & /@ celestialbodynames;
celestialbodies[[3, 2]] = AstronomicalData["Earth", "Radius"] Meter;
celestialbodies[[10, 2]] = 1000 AstronomicalData["Moon", "Apoapsis"] Meter;
ratioshand = (Fratio[#[[1]], 1 Kilogram, #[[2]], 6/100 Meter]) & /@ celestialbodies;
{celestialbodynames, ratioshand} // Transpose // MatrixForm
```

Out[27]= {Mercury, Venus, Earth, Mars, Jupiter, Saturn, Uranus, Neptune, Pluto, Moon}

Out[32]//MatrixForm=

$$\begin{pmatrix} \text{Mercury} & 0.024140 \\ \text{Venus} & 0.25723 \\ \text{Earth} & 5.2868\times 10^{8} \\ \text{Mars} & 0.014348 \\ \text{Jupiter} & 7.2924 \\ \text{Saturn} & 0.74622 \\ \text{Uranus} & 0.031260 \\ \text{Neptune} & 0.016829 \\ \text{Pluto} & 8.347\times 10^{-7} \\ \text{Moon} & 1607. \end{pmatrix}$$

Not much change. How about one more: a car passing you on the freeway.

In[33]:=
```
ratioscar = (Fratio[#[[1]], 1000 Kilogram, #[[2]], 2 Meter]) & /@ celestialbodies;
{celestialbodynames, ratioscar} // Transpose // MatrixForm
```

Out[34]//MatrixForm=

$$\begin{pmatrix} \text{Mercury} & 0.026822 \\ \text{Venus} & 0.28582 \\ \text{Earth} & 5.8742\times10^{8} \\ \text{Mars} & 0.015942 \\ \text{Jupiter} & 8.1027 \\ \text{Saturn} & 0.82914 \\ \text{Uranus} & 0.034733 \\ \text{Neptune} & 0.018699 \\ \text{Pluto} & 9.275\times10^{-7} \\ \text{Moon} & 1785. \end{pmatrix}$$

Jupiter still wins. OK, I can't resist, one more: you're visiting an aircraft carrier and standing on the dock looking up at it (from Wikipedia, the mass of an Essex class WWII aircraft carrier is 27200 tons and it's width is ~30m)

In[35]:=
```
ratioscarrier = (Fratio[#[[1]], Convert[27 200 × 2000 Pound, Kilogram], #[[2]], 50 Meter]) & /@
   celestialbodies;
{celestialbodynames, ratioscarrier} // Transpose // MatrixForm
```

Out[36]//MatrixForm=

$$\begin{pmatrix} \text{Mercury} & 0.00067937 \\ \text{Venus} & 0.0072394 \\ \text{Earth} & 1.48788\times10^{7} \\ \text{Mars} & 0.000403789 \\ \text{Jupiter} & 0.205231 \\ \text{Saturn} & 0.0210011 \\ \text{Uranus} & 0.000879753 \\ \text{Neptune} & 0.000473619 \\ \text{Pluto} & 2.34914\times10^{-8} \\ \text{Moon} & 45.2189 \end{pmatrix}$$

OK, we finally overcame Jupiter but only by a factor of five. If you're 110 meters from its center of gravity then

In[37]:=
```
ratioscarrier = (Fratio[#[[1]], Convert[27 200 × 2000 Pound, Kilogram], #[[2]], 110 Meter]) & /@
   celestialbodies;
{celestialbodynames, ratioscarrier} // Transpose // MatrixForm
```

Out[38]//MatrixForm=

$$\begin{pmatrix} \text{Mercury} & 0.00328815 \\ \text{Venus} & 0.0350387 \\ \text{Earth} & 7.20133\times10^{7} \\ \text{Mars} & 0.00195434 \\ \text{Jupiter} & 0.993319 \\ \text{Saturn} & 0.101646 \\ \text{Uranus} & 0.004258 \\ \text{Neptune} & 0.00229232 \\ \text{Pluto} & 1.13698\times10^{-7} \\ \text{Moon} & 218.859 \end{pmatrix}$$

Its gravitational force on you is equivalent to Jupiter's. Note also that we're closing in on the Moon. One might consider doing the calculation for the Sears Tower.

As a final observation, all this is relevent to why G, the gravitational constant, is the least well measured fundamental constant. There's so much mass around that it's hard to know where it all is.